**************CALC CUM ATTEMPTED HOURS AS OVERALL AND GET SOAPCOL****************

create or replace

FUNCTION F_FHDA_FA_SAP_CALC_STATS

        (p_pidm      number,

         p_calc_term varchar2,

         p_stat_type varchar2)

RETURN number AS

/* Whitelaw - SunGard Higher Education

   August 20, 2009

   Function to calculate stats to be used for SAP status determination

     TA - Term institutional attempted hours

     TE - Term institutional earned hours

     TG - Term institutional GPA

     CA - Cumulative institutional attempted hours

     CG - Cumulative institutional GPA

     CP - Cumulative institutional GPA as of prior term

  Stats must combine both FH and DA stats for given term and levels

  Testing SELECT

  select

    shrlgpa_pidm,

    f_fhda_fa_sap_calc_stats(shrlgpa_pidm,'200910','TA') TA,

    f_fhda_fa_sap_calc_stats(shrlgpa_pidm,'200910','TE') TE,

    f_fhda_fa_sap_calc_stats(shrlgpa_pidm,'200910','TG') TG,

    f_fhda_fa_sap_calc_stats(shrlgpa_pidm,'200910','CA') CA,

    f_fhda_fa_sap_calc_stats(shrlgpa_pidm,'200910','CG') CG,

    f_fhda_fa_sap_calc_stats(shrlgpa_pidm,'200910','CP') CP

  from shrlgpa 

*/

  v_result                     number;

  v_message                    varchar2(100);

  v_calc_term_fh               varchar2(6);

  v_calc_term_da               varchar2(6);

  v_term_hours_attempted       number;

  v_term_hours_earned          number;

  v_term_gpa_hours             number;

  v_term_quality_points        number;

  v_term_gpa                   number;

  v_cum_hours_attempted        number;

  v_cum_hours_earned           number;

  v_cum_gpa_hours              number;

  v_cum_quality_points         number;    

  v_cum_gpa                    number;

  v_cum_gpa_hours_prior        number;

  v_cum_quality_points_prior   number;    

  v_cum_gpa_prior              number;

BEGIN

----------------------------------------------------------------------

--Validate incoming term - should be a Foothill(FH) term

--Calculate corresponding De Anza(DA) term

----------------------------------------------------------------------

  begin

    select 

      stvterm_code

    into v_calc_term_fh

    from stvterm

    where stvterm_code = p_calc_term

      and substr(stvterm_code,5,1) = '1';

    exception

      when others then

        v_message := 'Invalid calc term';

        goto end_of_function;

  end;

  v_calc_term_da := substr(v_calc_term_fh,1,5) || '2';

----------------------------------------------------------------------

--Validate incoming stat type

----------------------------------------------------------------------

  if p_stat_type not in ('TA','TE','TG','CA','CG','CP') then

      v_message := 'Invalid stat type';

      goto end_of_function;

  end if;

----------------------------------------------------------------------

--Get stats

----------------------------------------------------------------------

  if p_stat_type like 'C%' then

      goto get_cum_stats;

  end if;

----------------------------------------------------------------------

--Get term stats

----------------------------------------------------------------------

  begin

    select

      sum(nvl(shrtgpa_hours_attempted,0)),

      sum(nvl(shrtgpa_hours_earned,0)),

      sum(nvl(shrtgpa_gpa_hours,0)),

      sum(nvl(shrtgpa_quality_points,0))

    into 

      v_term_hours_attempted,

      v_term_hours_earned,

      v_term_gpa_hours,

      v_term_quality_points

    from shrtgpa

    where shrtgpa_pidm = p_pidm

      and shrtgpa_term_code in (v_calc_term_fh,v_calc_term_da)

      and shrtgpa_levl_code in ('FH','DA')

      and shrtgpa_gpa_type_ind = 'I';

    exception

      when others then

        v_message := 'Error getting term stats';

        goto end_of_function;

  end;

  if v_term_gpa_hours = 0 then

      v_term_gpa := 0;

  else

      v_term_gpa := v_term_quality_points / v_term_gpa_hours;

  end if;

  goto end_of_function;

----------------------------------------------------------------------

--Get cum stats

----------------------------------------------------------------------

<<get_cum_stats>>

  begin

    select

      sum(nvl(shrlgpa_hours_attempted,0)),

      sum(nvl(shrlgpa_hours_earned,0)),

      sum(nvl(shrlgpa_gpa_hours,0)),

      sum(nvl(shrlgpa_quality_points,0))

    into 

      v_cum_hours_attempted,

      v_cum_hours_earned,

      v_cum_gpa_hours,

      v_cum_quality_points

    from shrlgpa

    where shrlgpa_pidm = p_pidm

      and shrlgpa_levl_code in ('FH','DA')

      and shrlgpa_gpa_type_ind = 'I';

    exception

      when others then

        v_message := 'Error getting cum stats';

        goto end_of_function;

  end;

  if v_cum_gpa_hours = 0 then

      v_cum_gpa := 0;

  else

      v_cum_gpa := v_cum_quality_points / v_cum_gpa_hours;

  end if;

  if p_stat_type <> 'CP' then

      goto end_of_function;

  end if;

----------------------------------------------------------------------

--Get cum gpa as of prior term

----------------------------------------------------------------------  

  begin

    select

      sum(nvl(shrtgpa_gpa_hours,0)),

      sum(nvl(shrtgpa_quality_points,0))

    into 

      v_cum_gpa_hours_prior,

      v_cum_quality_points_prior

    from shrtgpa

    where shrtgpa_pidm = p_pidm

      and shrtgpa_term_code < v_calc_term_fh

      and shrtgpa_levl_code in ('FH','DA')

      and shrtgpa_gpa_type_ind = 'I';

    exception

      when others then

        v_message := 'Error getting cum stats';

        goto end_of_function;

  end;

  if v_cum_gpa_hours_prior = 0 then

      v_cum_gpa_prior := 0;

  else

      v_cum_gpa_prior := v_cum_quality_points_prior / v_cum_gpa_hours_prior;

  end if;

  goto end_of_function;

----------------------------------------------------------------------

--End of function - return result

----------------------------------------------------------------------

<<end_of_function>>  

  if v_message is not null then

      return null;

  else

      v_result := 

        case 

          when p_stat_type = 'TA' then v_term_hours_attempted

          when p_stat_type = 'TE' then v_term_hours_earned

          when p_stat_type = 'TG' then v_term_gpa

          when p_stat_type = 'CA' then v_cum_hours_attempted

          when p_stat_type = 'CG' then v_cum_gpa

          when p_stat_type = 'CP' then v_cum_gpa_prior

          else null

        end;

      return v_result;

  end if;

END F_FHDA_FA_SAP_CALC_STATS;

